

Traiettorie di Sonde ad Accelerazione Tangenziale Costante risolte con le Unità di misura Canoniche (UC)

Giuseppe Matarazzo

Giugno 2004

Sommario

Questo studio teorico mostra alcune traiettorie di veicoli spaziali a cui è impressa un'accelerazione costante secondo la direzione di volo. Si tratta di spirali generate da una debole accelerazione (centesimi di m/s^2) e di curve in cui essa è più forte e che in poco tempo 'schizzano via' dal corpo gravitazionale attorno a cui ruotavano, di moto circolare uniforme, prima dell'innesco della spinta. L'uso di unità di misura normalizzate (Unità Canoniche) consente di applicare gli algoritmi di integrazione numerica a qualsiasi moto gravitazione centrale.

1 Introduzione

Nella letteratura astronomica si fa spesso riferimento a sonde sottoposte a debole o forte **spinta** diretta nella direzione del moto e quasi mai vengono forniti dettagli sul calcolo delle loro traiettorie. Nel caso di *thrust* notevoli, con consumo consistente di propellente, la variazione di massa $\frac{dm}{dt}$ è tutt'altro che trascurabile per cui ogni traiettoria è legata ad uno specifico velivolo, avente una massa iniziale ben definita; l'analisi, in questi casi, non può essere generalizzata.

Se invece ammettiamo che siano le *accelerazioni* ad essere costanti, ovvero che sia ininfluenza la perdita di massa della sonda, allora il problema può essere affrontato nella sua globalità, tenendo presente che per forti spinte unitarie esso deve restare confinato nell'ambito teorico.

2 Le Unità Canoniche

In astronomia le unità di misura *MKS* (metro, chilogrammo, secondo) del Sistema Internazionale sono tutt'altro che comode da usare per cui si adotta, per esempio nel sistema solare, come misura delle lunghezze l'*unità astronomica* UA, pari alla distanza media tra Sole e Terra. E c'è di più. Per far sì che il parametro gravitazionale del Sole $\mu = G \cdot M$ fosse *unitario*, ovvero $\mu = 1$, Gauss 'inventò' la sua *costante* gravitazionale $k = 0.017\ 202\ 098\ 95$, che rappresenta l'**unità canonica** di velocità in UA/giorno.

Nel nostro caso, le grandezze da rendere unitarie sono DUE: il solito parametro gravitazionale μ e il raggio dell'orbita di parcheggio r_0 del veicolo spaziale. Si suppone che questo ruoti, prima dell'azione della spinta tangenziale costante, di moto circolare uniforme di raggio r_0 intorno al corpo massivo (Terra o pianeta qualsiasi) di parametro μ .

In tal caso le *altre* unità di misura risultano:

Tabella delle UC

<i>Tempo</i>	$\sqrt{\frac{r_0^3}{\mu}}$
<i>Raggio Vettore</i>	r_0
<i>Velocita'</i>	$\sqrt{\frac{\mu}{r_0}}$
<i>Accelerazione</i>	$\frac{\mu}{r_0^2}$
<i>Sviluppo traiettoria</i>	r_0
<i>Anomalia θ</i>	<i>radianti</i>

3 Le 3 spirali a debole spinta

Nelle prime tre figure che si trovano in fondo a questa relazione sono state rappresentate, in coordinate polari (r_0, θ) , le rispettive traiettorie con *deboli* accelerazioni. E' stata considerata una sonda *ideale* che si muove nell'ambito dell'attrazione gravitazionale terrestre ($\mu = 398\,600 \text{ km}^3/\text{s}^2$) con raggio iniziale r_0 pari a quello terrestre $R = 6378 \text{ km}$ a cui vengono impresse le seguenti tre *accelerazioni*.

0.001	m/s^2	=	$0.001 \cdot \frac{\mu}{r_0^2}$	=	0.000 102 04	UC
0.01	m/s^2	=	$0.01 \cdot \frac{\mu}{r_0^2}$	=	0.001 020 4	UC
0.05	m/s^2	=	$0.05 \cdot \frac{\mu}{r_0^2}$	=	0.051 020	UC

Il programma di integrazione numerica, che verrà integralmente trascritto più avanti (listato Pascal), consente di calcolare TUTTI i parametri delle curve, dopo aver scelto un opportuno intervallo di integrazione dt , espresso sempre in UC.

Nel caso della spirale avente $a=0.01 \text{ m/s}^2$ il file risultati si presenta così (parte finale della traiettoria):

tempo UC Radq(r_0^3/μ)	anom. (gra)	Rag_Vett. UC r_0		Veloc. UC Radq(μ/r_0)		Accel. UC μ/r_0^2		Sviluppo UC r_0
810.5	14065.3976	23.08394		0.267102		0.0010204		482.50851
810.6	14065.4532	23.09852		0.267102		0.0010204		482.53522
810.7	14065.5087	23.11311		0.267102		0.0010204		482.56193
810.8	14065.5641	23.12770		0.267102		0.0010204		482.58864
810.9	14065.6195	23.14230		0.267102		0.0010204		482.61535
811.0	14065.6748	23.15691		0.267101		0.0010204		482.64206
811.1	14065.7301	23.17153		0.267102		0.0010204		482.66877
811.2	14065.7854	23.18616		0.267102		0.0010204		482.69548
811.3	14065.8406	23.20080		0.267102		0.0010204		482.72219
811.4	14065.8957	23.21544		0.267102		0.0010204		482.74890
811.5	14065.9508	23.23010		0.267102		0.0010204		482.77561

Da notare, nella parte ‘staccata’, il punto di velocità *minima*, oltre il quale il veicolo sentirà sempre meno l’attrazione gravitazionale del corpo massivo da cui è partito.

Questo punto di inversione si verifica quando, dall’inizio della spinta, è passato il tempo canonico 811.0, pari a $811.0 \cdot \sqrt{\frac{r_0^3}{\mu}} = 811.0 \cdot 806.78$ secondi = 7.57 giorni. A quell’epoca la sonda ha effettuato $14065.6748/360 = 39.07$ rivoluzioni e percorso una traiettoria di $482.64206 \cdot 6378 = 3\,078\,291$ km. Il suo raggio vettore è di $23.15691 \cdot 6378 = 147\,694$ km e la sua velocità di $0.267101 \cdot \sqrt{\frac{\mu}{r_0}} = 0.267101 \cdot 7.905446 = 2.11155$ km/s.

4 Le traiettorie a forte spinta

Ne sono state calcolate 5, con accelerazioni che variano da 0.2 a 1.0 UC, e raggruppate nell’ultima figura riportata in calce a questo lavoro.

5 Listato Pascal del Programma

```
Program UC;      { Traiettorie Accelerate risolte usando le Unità di misura
                  Canoniche (UC) - 17/18.06.2004 - Integrazione numerica
                  effettuata con l'algoritmo Runge-Kutta del 5.o ordine
                  di Antonino Buto'.
                  In fase esecutiva il programma si blocca premendo un
                  tasto qualsiasi }
```

```
{ $N+ }
```

```
Uses Crt;
```

```
Var
```

```
  i: Integer;
  vv,vc0,vta,vra,rad,dt,gama,rv,AA,Sv : Real;
  teta,gama0,vr0,vv0,ra : Real;
  tmp,dur : Double;
  f,df,f0,h1,h2,h3,h4,h5,h6 : Array[1..6] of Double;
  g : Text;
```

```
Procedure Iter;
```

```
  Begin
    df[1]:=f[3];          (* dr/dt=Vr *)
    df[2]:=f[4]/f[1];    (* dteta/dt=Vt/r *)
    vv:=sqrt(sqr(f[3])+sqr(f[4]));
    rv:=f[1];
    df[3]:=-1/sqr(f[1])+AA*f[3]/vv+sqr(f[4])/f[1];
    df[4]:=-f[4]*f[3]/f[1]+AA*f[4]/vv;
  End;
```

```
Procedure Runge;
```

```
  Begin
  for i:=1 to 4 do
    begin
      f[i]:=f0[i];
    end;
```

```

Iter;
for i:=1 to 4 do
begin
  h1[i]:=dt*df[i];
  f[i] :=f0[i]+h1[i]/4;
end;

Iter;
for i:=1 to 4 do
begin
  h2[i]:=dt*df[i];
  f[i] :=f0[i]+h1[i]/8+h2[i]/8;
end;

Iter;
for i:=1 to 4 do
begin
  h3[i]:=dt*df[i];
  f[i] :=f0[i]-h2[i]/2+h3[i];
end;

Iter;
for i:=1 to 4 do
begin
  h4[i]:=dt*df[i];
  f[i] :=f0[i]+3*h1[i]/16+9*h4[i]/16;
end;

Iter;
for i:=1 to 4 do
begin
  h5[i]:=dt*df[i];
  f[i] :=f0[i]-3*h1[i]/7+2*h2[i]/7+12*h3[i]/7;
  f[i] :=f[i]-12*h4[i]/7+8*h5[i]/7;
end;

Iter;
for i:=1 to 4 do
begin
  h6[i]:=dt*df[i];
  f[i] :=f0[i]+(7*h1[i]+32*h3[i]+12*h4[i]+32*h5[i]+7*h6[i])/90;
  f0[i]:=f[i];
end;

End;

Procedure Scrivi;
Begin
  writeln(tmp:8:1,' ',teta/rad:13:4,' ',rv:10:5,' | ',vv:9:6,

```

```

' | ',AA:9:7,' | ',Sv:9:5);
writeln(g,tmp:8:1,' ',teta/rad:13:4,' ',rv:10:5,' | ',vv:9:6,
' | ',AA:9:7,' | ',Sv:9:5);
End;

```

```

Procedure Main;

```

```

Begin

```

```

repeat

```

```

tmp:=tmp+dt;

```

```

Runge;

```

```

teta:=f[2];

```

```

gama:=arctan(f[3]/f[4]);

```

```

rv:=f[1];

```

```

vv:=sqrt(sqr(f[3])+sqr(f[4]));

```

```

ra:=rv;

```

```

vta:=f[4];

```

```

vra:=f[3];

```

```

{ Sviluppo curva:  $s = (V^2 - \mu * (2/r - 1/r_0)) / (2 * AA)$  }

```

```

Sv:=(sqr(vv)-(2/rv-1))/(2*AA);

```

```

Scrivi;

```

```

until keypressed;

```

```

End;

```

```

Procedure Blocco;

```

```

Begin

```

```

tmp:=0;

```

```

ra:=1;

```

```

vr0:=0;

```

```

vc0:=1;

```

```

vta:=vc0;

```

```

vra:=vr0;

```

```

gama:=0;

```

```

vv:=vc0;

```

```

gama0:=gama;

```

```

vv0:=vv;

```

```

f0[1]:=1; (* r *)

```

```

f0[2]:=0; (* teta *)

```

```

f0[3]:=vv*sin(gama); (* dr/dt = Vr *)

```

```

f0[4]:=vv*cos(gama); (* Vt = r*dteta/dt *)

```

```

tmp:=0;

```

```

teta:=0;

```

```

rv:=1;

```

```

f[3]:=f0[3];

```

```

f[4]:=f0[4];
Scrivi;

Main;
End;

{----- INIZIO PROGRAMMA -----}
Begin
  TextBackGround(9);
  TextColor(15);
  writeln;
  writeln(' ..... attendere file finale: OUTPUT.RK5');

  rad:=pi/180;
  tmp:=0;

  {===== INIZIO dati =====}
  dt:=1;          { dt= variabile di integrazione }
  AA:=0.00010204; { 1 UC-accel.= (mu/r0^2) }
  {===== FINE dati =====}

  Assign(g,'output.rk5'); { File OUTPUT }
  Rewrite(g);
  writeln(g,'-----+-----+-----+-----');
  writeln(g,'   tempo      anom.  Rag_Vett.|  Veloc.  |  Accel.  |  Sviluppo  ');
  writeln(g,'   UC          (gra)   UC    |   UC    |   UC    |   UC    ');
  writeln(g,' Radq(r0^3/mu)          r0    | Radq(mu/r0)| mu/r0^2 |   r0    ');
  writeln(g,'-----+-----+-----+-----');

  rv:=1; {inizializzazione raggio vettore: rv= r0=1}

  Blocco;
  writeln;
  writeln(' ..... Risultati sul file: OUTPUT.RK5');
  Close(g);
End.

```

6 Conclusione

La versatilità delle UC si evidenzia maggiormente quando si devono calcolare le traiettorie di *sonde accelerate* che partono da qualsiasi pianeta, non necessariamente dalla Terra. Infatti a parità di accelerazioni **canoniche**, indicate nel listato con la grandezza AA, i risultati che si ottengono, ossia tempi, velocità, anomalie, raggi vettori e lunghezze delle curve, sono **uguali** per qualsiasi corpo. Il parametro μ e il raggio dell'orbita iniziale r_0 ne differenziano le grandezze assolute.

..... 25.06.2004

Sonda sottoposta a DEBOLE Accelerazione Tangenziale Costante: Unità Canoniche (UC)




Figura 1: Traiettoria sonda con $a= 0.001 \text{ m/s}^2$




Figura 2: Traiettoria sonda con $a= 0.01 \text{ m/s}^2$




Figura 3: Traiettoria sonda con $a = 0.05 \text{ m/s}^2$




Figura 4: Traiettorie sonde con $a= 1.0$ -nero, 0.8 -blu, 0.6 -rosso, 0.4 -verde, 0.2 -viola UC